Supplementary Material 1: Computer Programs

(Novak & Tyson, Design Principles of Biochemical Oscillators)

Purpose: to provide 'ode' files for simulating all models in the text, using WinPP or XPP-Aut. These programs are freely available at
http://www.math.pitt.edu/~bard/bardware/
http://www.math.pitt.edu/~bard/xpp/xpp.html
.

Fig.1C-D

```
# protein inhibits its own synthesis with explicit time delay
# protein is degraded by a protease according to Michaelis-Menten kinetics

dy/dt = k1*S*Kd^p/(Kd^p + delay(y,tau)^p) - k2*ET*y/(Km + y)

aux dly = delay(y,tau)
aux Fy=k1*S*Kd^p/(Kd^p + delay(y,tau)^p)

p k1=1, S=1, Kd=1, p=2, tau=10
p k2=1, ET=1, Km=1

@ XP=t, YP=y, TOTAL=100, METH=stiff, XLO=0, XHI=100, YLO=0, YHI=3.5,
delay=20
done
```

Fig.2B

```
# model for a simple negative feedback loop
# protein (y) inhibits the synthesis of its mRNA (x)

dx/dt = k1*S*Kd^p/(Kd^p + y^p) - kdx*x
dy/dt = ksy*x - k2*ET*y/(Km + y)

p k1=0.1, S=1, Kd=1, kdx=0.1, p=2
p ksy=1, k2=1, ET=1, Km=1

@ XP=y, YP=x, TOTAL=100, METH=stiff, XLO=0, XHI=4, YLO=0, YHI=1.05
done
```

Fig.2D-F

```
# Negative feedback loop with nuclear transport
# mRNA is synthesized in the nucleus (xn) and transported into the cytoplasm (xc)
# where it gets translated into protein (yc) which is tranlocated into the nucleus (yn)
#
# eps = Vnuc/Vcyt
# half-life of mRNA in nucleus = 0.693/kdxn
# half-life of prot in cytoplasm = 0.693/kdyc

dxn/dt = kdxn*(sig/(1 + yn^p) - xn) - kexport*xn
dxc/dt = eps*kexport*xn - kdxc*xc
dyc/dt = kdyc*(xc - yc)  - eps*kimport*yc
dyn/dt = kimport*yc - kdyn*yn/(Km + yn)

p Sig=1000, p=2, kdxn=10, kexport=0.2,kdxc=0.2, eps=1
p kdyn=8, kdyc=0.1, Km=0.1, kimport=0.1

@ XP=t, YP=xn, TOTAL=100, METH=stiff, XLO=0, XHI=100, YLO=0, YHI=1000,
bounds=10000
done
```

Fig.3B-F

```
# protein inhibits its own degradation

dx/dt = k1*S*Kd^p/(Kd^p + y^p) - kdx*x
dy/dt = ksy*x - kdy*y - k2*ET*y/(Km + y + KI*y^2)

p S=1, k1=0.05, Kd=1, p=4, kdx=0.05
p ksy=1, kdy=0.05, k2=1, ET=1, Km=0.1, KI=2

@ XP=y, YP=x, TOTAL=250, METH=stiff, XLO=0, XHI=4, YLO=0, YHI=1.05
done
```

Fig.5A (left)

```
# Three component negative feedback oscillator (+ + -)
# X -> Y -> Z -| X

dx/dt = ksx*S/(1 + z^p) - kdx*x
dy/dt = k1*x - k2*y/(Km + y)
dz/dt = k3*(y - z)

p ksx=0.1, S=2, kdx=0.1, p=4
p k1=0.2, k2=0.1, Km=0.01, k3=0.05

@ XP=t, YP=x, TOTAL=250, METH=stiff, XLO=0, XHI=250, YLO=0, YHI=1
done
```

Fig.5B (left)

```
# Activator amplification & negative feedback

dx/dt = ksx' + ksx*w - (kdx' + kdx*y)*x
dy/dt = ksy*x^p/(1 + x^p) - kdy*y

w = x^q/(1 + x^q)

p ksx'=0.02, ksx=1, kdx'=0.2, kdx=1, q=2
p ksy=0.01, kdy=0.01, p=2

@ XP=y, YP=x, TOTAL=250, METH=stiff, XLO=0, XHI=0.5, YLO=0, YHI=5
done
```

Fig.5B (right)

```
# Inhibitor amplification with negative feedback
# x promotes y synthesis and y promotes x degradation
# y degradation is enhanced by z
# y binds to z to form an inactive complex C
#
# Kdiss*C = Yfree*Zfree = (Ytotal-C)*(Ztotal-C)

dx/dt = ksx - (kdx' + kdx*y^p)*x
dy/dt = ksy' + ksy*x - (kdy' + kdy*z)*Y


BB = ZT+Y+Kdiss
CC = 2*ZT*Y/(BB+sqrt(BB^2-4*ZT*Y))
```

```
z = ZT - CC

p ksx=0.01, kdx'=0.01, kdx=1, p=2
p ksy'=0.1, ksy=0.2, kdy'=0.1, kdy=250, Kdiss=0.01
p ZT=0.05

@ XP=y, YP=x, TOTAL=250, METH=stiff, XLO=0, XHI=0.5, YLO=0, YHI=1
done
```

Fig.6B-C

```
# Rossler (1977) BMB
#
# Spiral Chaos

dx/dt = k1 + k2*x - (k3*y+k4*z)*x/(K+x)
dy/dt = k5*x - k6*y
dz/dt = k7*x - k8*z/(L+z)

p k1=22, k2=2.2, k3=4.4, k4=4.4, k5=1.2
p k6=1, k7=14, k8=140, K=0.01, L=0.05

init x=7, y=6, z=0.1
done
```

Supplementary Material 2: Negative Feedback with Explicit Time Delay

(Novak & Tyson, Design Principles of Biochemical Oscillators)

Purpose: to derive the constraint between τ and S for oscillatory solutions to Eq. (2) of the text. The constraint equation is used to plot the curves in Fig. 1e and f.

First, we write Eq. (2) in dimensionless form:

$$\frac{dy(\hat{t})}{d\hat{t}} = \frac{\sigma}{1+\left[y(\hat{t}-\hat{\tau})\right]^p} - \frac{y}{\kappa+y} \text{ , where } y(\hat{t}) = \frac{Y(t)}{K_d} \text{ and } \hat{t} = \frac{k_2 E_T}{K_d} t \text{ .} \quad \text{(S2.1)}$$

The parameters in this equation are defined by $\sigma = \frac{k_1 S}{k_2 E_T}$, $\kappa = \frac{K_m}{K_d}$, $\hat{\tau} = \frac{k_2 E_T}{K_d} \tau$.

The steady state solution of Eq. (S2.1) is the unique real positive root, y_o, of

$$y^{p+1} - (\sigma-1)y - \sigma\kappa = 0 \text{ .} \quad \text{(S2.2)}$$

Assume $y(\hat{t}) = y_o + ce^{i\omega\hat{t}}$, where c = constant. Substituting this assumed solution into Eq. (S2.1), we find that

$$i\omega = \rho - \phi e^{-i\omega\hat{\tau}} \quad \text{(S2.3)}$$

where $\rho = \frac{\kappa}{(\kappa+y_o)^2}$ and $\phi = \frac{\sigma p y_o^{p-1}}{(1+y_o^p)^2}$. Using (S2.2) we find a convenient relation between ϕ and ρ: $\frac{\phi}{\rho} = p\left[1+(\sigma-1)\frac{y_o}{\sigma\kappa}\right] \equiv R$, where R is a label for $p[\ldots]$.

Equating the real and imaginary parts of Eq. (S2.3), we find that

$$\rho^2 + \omega^2 = \phi^2 \text{ and } \omega\hat{\tau} = \arctan\left(-\omega/\rho\right) \text{.} \quad \text{(S2.4)}$$

The conditions (S2.4) imply that

$$\omega = \rho\sqrt{R^2-1} \text{ and } \hat{\tau} = \omega^{-1}\arctan\left(-\sqrt{R^2-1}\right) \text{.} \quad \text{(S2.5)}$$

We are now prepared to compute the curves in Fig. 1e and f, by the following pseudocode:

```
Scan over p=1, 2, 3, …
     Scan over values of κ
          Scan over values of σ
               Solve (S2.2) for yo
               Compute R=p*[1+(σ-1)*yo/(σκ)]
               If (R>1), then
                    Compute ρ=κ/(κ+yo)^2
                    Compute ω=ρ*sqrt(R*R-1)
                    Compute τ=arctan(-ω/ρ)/ω
                    Compute Period=2*π/ω
                    Save (σ,τ,ω,Period)
               Else continue
          Continue
     For each (p,κ),plot (τ versus σ)
     Continue
Continue
End
```

Supplementary Material 3: Motif G

(Novak & Tyson, Design Principles of Biochemical Oscillators)

Purpose: to prove that the incoherently amplified NFL in motif (G), with a 2-component negative feedback loop and a 3-component positive feedback loop, cannot generate oscillations by a Hopf bifurcation.

For motif (G) proposed in the text (also below, left), we exhibit the sign pattern of the Jacobian matrix (below, right), where the *a*'s, *b*'s and *c*'s are all > 0,



$$\mathbf{J} = \begin{pmatrix} -a_x & -a_y & 0 \\ b_x & -b_y & \mp b_z \\ \pm c_x & 0 & -c_z \end{pmatrix}$$

The stability of the steady state depends on the eigenvalues, λ, of the Jacobian matrix, which are the roots of the characteristic equation:

$$0 = \det \begin{pmatrix} -a_x - \lambda & -a_y & 0 \\ b_x & -b_y - \lambda & \mp b_z \\ \pm c_x & 0 & -c_z - \lambda \end{pmatrix}$$

$$0 = (\lambda + a_x)(\lambda + b_y)(\lambda + c_z) + a_y b_x (\lambda + c_z) - a_y b_z c_x \qquad \text{(S3.1)}$$

$$0 = \lambda^3 + \lambda^2 \left(a_x + b_y + c_z\right) + \lambda\left(a_x b_y + b_y c_z + a_x c_z + a_y b_x\right) + a_x b_y c_z + a_y b_x c_z - a_y b_z c_x$$

In order for a Hopf bifurcation to occur, this algebraic equation must have a pair of pure imaginary roots, $\lambda = \pm i\omega$. The necessary and sufficient condition for pure imaginary roots to Eq. (S3.1) is

$$a_x b_y c_z + a_y b_x c_z - a_y b_z c_x = \left(a_x + b_y + c_z\right)\left(a_x b_y + b_y c_z + a_x c_z + a_y b_x\right)$$

$$0 = 2a_x b_y c_z + a_x\left(a_x b_y + a_x c_z + a_y b_x\right) + b_y\left(a_x b_y + b_y c_z + a_y b_x\right) + c_z\left(b_y c_z + a_x c_z\right) + a_y b_z c_x \qquad \text{(S3.2)}$$

Clearly, Eq. (S3.2) cannot be satisfied for any choice of a_x, etc. Hence, it is impossible for this motif to generate oscillations by a Hopf bifurcation. On the other hand, λ = 0 is a possible solution of Eq. (S3.1), if $a_x b_y c_z + a_y b_x c_z = a_y b_z c_x$. Hence, this motif can generate multiple steady states by saddle-node bifurcations. So we conclude that motif (G) can exhibit bistability but not oscillations.

By a similar argument, we can come to the same conclusion for motif (G') below



$$\mathbf{J} = \begin{pmatrix} -a_x & -a_y & \pm a_z \\ b_x & -b_y & 0 \\ 0 & \pm c_y & -c_z \end{pmatrix}$$

Supplementary Material 4: Motif H

(Novak & Tyson, Design Principles of Biochemical Oscillators)

Purpose: to prove that motif (H) (below, left), for which both the 2-component and 3-component feedback loops are positive, can exhibit bistability but cannot generate stable oscillations by a Hopf bifurcation.



$$\mathbf{J} = \begin{pmatrix} -a_x & a_y & \pm a_z \\ b_x & -b_y & 0 \\ 0 & \pm c_y & -c_z \end{pmatrix}$$

The sign pattern of the Jacobian matrix for motif (H) is given above (right), where the *a*'s, *b*'s and *c*'s are all > 0.The stability of the steady state depends on the eigenvalues, λ, of the Jacobian matrix, which are the roots of the characteristic equation:

$$0 = \det \begin{pmatrix} -a_x - \lambda & a_y & \mp a_z \\ b_x & -b_y - \lambda & 0 \\ 0 & \mp c_y & -c_z - \lambda \end{pmatrix}$$

$$0 = (\lambda + a_x)(\lambda + b_y)(\lambda + c_z) - a_y b_x (\lambda + c_z) - a_z b_x c_y \qquad \text{(S4.1)}$$

$$0 = \lambda^3 + \lambda^2 \left(a_x + b_y + c_z\right) + \lambda\left(a_x b_y + b_y c_z + a_x c_z - a_y b_x\right) + a_x b_y c_z - a_y b_x c_z - a_z b_x c_y$$

In order for a Hopf bifurcation to occur, this algebraic equation must have a pair of pure imaginary roots, $\lambda = \pm i\omega$. The necessary and sufficient condition for pure imaginary roots to Eq. (S4.1) is

$$a_x b_y c_z - a_y b_x c_z - a_z b_x c_y = \left(a_x + b_y + c_z\right)\left(a_x b_y + b_y c_z + a_x c_z - a_y b_x\right)$$

$$a_y b_x (a_x + b_y) - a_z b_x c_y = 2 a_x b_y c_z + a_x^2 \left(b_y + c_z\right) + b_y^2 \left(a_x + c_z\right) + c_z^2 \left(a_x + b_y\right) \qquad \text{(S4.2)}$$

$$0 = 2 a_x b_y c_z + c_z \left(a_x^2 + b_y^2\right) + c_z^2 \left(a_x + b_y\right) + a_z b_x c_y + \left(a_x b_y - a_y b_x\right)(a_x + b_y)$$

If $(a_x b_y - a_y b_x) > 0$, then Eq. (S4.2) cannot be satisfied for any choice of a_x, etc. Hence, if motif (H) is to generate limit cycle oscillations by a Hopf bifurcation, then $(a_x b_y - a_y b_x)$ must be < 0. But, in that case, the characteristic equation (S4.1) must have a real positive root, $\lambda_1 > 0$, as well as a pair of pure imaginary eigenvalues. The bifurcating limit cycles must be unstable. We conclude that it is impossible for motif (H) to generate stable oscillations by a Hopf bifurcation.

On the other hand, λ = 0 is a possible solution of Eq. (S4.1), if $a_x b_y c_z = a_z b_x c_y + a_y b_x c_z$. Hence, motif (H) can generate multiple steady states by saddle-node bifurcations.

By a similar argument, we can come to the same conclusion for motif (H') below:



$$\mathbf{J} = \begin{pmatrix} -a_x & -a_y & 0 \\ -b_x & -b_y & \mp b_z \\ \mp c_x & 0 & -c_z \end{pmatrix}$$